

4.12 Isoparametrische räumliche Elemente

Das *isoparametrische Konzept* ist ausführlich in Ziffer 4.7 unter Beschränkung auf ebene Elemente behandelt worden. Darüber hinaus findet man in Ziffer 4.8 Hinweise, wie man *konforme Abbildungen* im *isoparametrischen* bzw. *subparametrischen Konzept* der FEM einsetzen kann.

Die Güte des isoparametrischen Konzeptes kann sehr einfach am Beispiel einer isoparametrischen Abbildung des Master-Quadrates auf den Einheitskreis getestet werden.

Dazu genügt es, die Seite $\xi = 1$ des biquadratischen Master-Quadrates vom LAGRANGE- oder SERENDIPITY-Typ (Bild 4.22) mit den Knotenpunkten ②, ⑥, ③ näherungsweise unter Beachtung von (4.138a,b) und (4.140) auf ein Viertel des Einheitskreises abzubilden:

$$x(1, \eta) = N_2(1, \eta)x_2 + N_3(1, \eta)x_3 + N_6(1, \eta)x_6, \quad (4.274a)$$

$$y(1, \eta) = N_2(1, \eta)y_2 + N_3(1, \eta)y_3 + N_6(1, \eta)y_6. \quad (4.274b)$$

Die Formfunktionen erhält man aus (4.129b,c,f) bzw. (4.133a,c) zu

$$N_2(1, \eta) = -\frac{1}{2}(1-\eta)\eta, \quad N_3(1, \eta) = \frac{1}{2}(1+\eta)\eta, \quad N_6(1, \eta) = 1-\eta^2. \quad (4.275a,b,c)$$

Die Punkte ②, ⑥, ③ auf dem Einheitskreis haben die Koordinaten $P_2(1, 0)$, $P_3(0, 1)$, $P_6(1/\sqrt{2}, 1/\sqrt{2})$, so dass (4.274a,b) mit (4.275a,b,c) in die Abbildung

$$x(1, \eta) = -\frac{1}{2}(1-\eta)\eta + \frac{1}{2}\sqrt{2}(1-\eta^2) \quad (4.276a)$$

$$y(1, \eta) = \frac{1}{2}(1+\eta)\eta + \frac{1}{2}\sqrt{2}(1-\eta^2) \quad (4.276b)$$

übergeht, die im Folgenden MAPLE -Programm dargestellt ist (Bild 4.52).

```

>
> x(1,eta):=-1/2*(1-eta)*eta+(1-eta^2)/sqrt(2);
          x(1, η) := -\frac{(1-\eta)\eta}{2} + \frac{(1-\eta^2)\sqrt{2}}{2}
> y(1,eta):=1/2*(1+eta)*eta+(1-eta^2)/sqrt(2);
          y(1, η) := \frac{(1+\eta)\eta}{2} + \frac{(1-\eta^2)\sqrt{2}}{2}
> plot1:=plot([x(1,eta),y(1,eta),eta=1..1],scaling=constrained,thickness=2):
> plot2:=plot(sqrt(1-x^2),x=0..1,color=black,scaling=constrained):
> plots[display]({plot1,plot2}):
>
    
```




Bild 4.52 Isoparametrische Abbildung des Master-Quadrates auf den Einheitskreis

Zu den Aufgaben der *Approximationstheorie* gehört es, eine durch einen analytischen Ausdruck definierte Funktion $y(x)$, hier der Einheitskreis $y(x) = \sqrt{1-x^2}$ bzw. $r(\eta) = 1$, in einem bestimmten Intervall durch eine *Ersatzfunktion* $\tilde{y}(x)$, hier die isoparametrische Abbildung $r(\eta)$, optimal zu approximieren. Zur Lösung dieser Aufgabe kann die L_2 -Fehlernorm minimiert werden:

$$\|\tilde{y}(x) - y(x)\|_2 := \sqrt{\int_0^1 [\tilde{y}(x) - y(x)]^2 dx} \stackrel{!}{=} \text{Minimum} \tag{4.277}$$

Die Güte der Approximation in Bild 4.52 kann an der L_2 -Fehlernorm

$$L_2 = \sqrt{\int_{-1}^1 [r(\eta) - 1]^2 d\eta} \quad (4.278)$$

gemessen werden, die mit Hilfe des folgenden MAPLE-Programms zu $L_2 = 0,0097 < 1\%$ bestimmt wird.

```
>
> x:=-1/2*(1-eta)*eta+(1-eta^2)/sqrt(2):
> y:=1/2*(1+eta)*eta+(1-eta^2)/sqrt(2):
> rr:=x^2+y^2:
> rr:=expand(%):
> rr:=factor(%):
> r:=+sqrt(%):
> L_2:=evalf(sqrt(int((1-r)^2,eta=-1..1)));
```

⊙ 4.12-2.mws

$L_{-2} := 0.009728103616$

```
> plot(1-r,eta=-1..1):
>
```

Darin werden die einzelnen Rechenschritte ausgedruckt, wenn man die Doppelpunkte durch Semikola ersetzt.

Es wäre sinnvoll, das Integral unterhalb der Wurzel in (4.278) auf die Länge des Integrationsintervalls zu beziehen. Damit würde man einen L_2 -Wert von $L_2 = 0.0068787$ erhalten, der um den Faktor $1/\sqrt{2}$ kleiner ist als der im MAPLE-Programm ausgewiesene Wert.

Zur Verbesserung der Approximation können kubische Formfunktionen der SERENDIPITY-Klasse gewählt werden. Die Seite $\xi = 1$ in Bild 4.24 wird isoparametrisch auf eine verbesserte Näherung abgebildet, die mit dem Einheitskreis vier Punkte gemeinsam hat: $P_2(1, 0)$, $P_3(0, 1)$, $P_7\left(\cos\frac{\pi}{6}, \sin\frac{\pi}{6}\right)$, $P_8\left(\cos\frac{\pi}{3}, \sin\frac{\pi}{3}\right)$.

Analog (4.275a,b,c) erhält man aus (4.134a,b) folgende Formfunktionen

$$N_2(1, \eta) = -\frac{1}{16}(1-\eta)(1-9\eta^2), \quad N_3(1, \eta) = -\frac{1}{16}(1+\eta)(1-9\eta^2), \quad (4.279a,b)$$

$$N_7(1, \eta) = \frac{9}{16}(1-\eta^2)(1-3\eta), \quad N_8(1, \eta) = \frac{9}{16}(1-\eta^2)(1+3\eta) \quad (4.279c,d)$$

und damit unter Berücksichtigung der Koordinaten (x_i, y_i) der Punkte P_2 , P_3 , P_7 , P_8 schließlich die Abbildung

$$x(1, \eta) = -\frac{1}{16}(1-\eta)(1-9\eta^2) + \frac{9}{16}(1-\eta^2)(1-3\eta) \cos \frac{\pi}{6} + \frac{9}{16}(1-\eta^2)(1+3\eta) \cos \frac{\pi}{3} \quad (4.280a)$$

$$y(1, \eta) = -\frac{1}{16}(1+\eta)(1-9\eta^2) + \frac{9}{16}(1-\eta^2)(1-3\eta) \sin \frac{\pi}{6} + \frac{9}{16}(1-\eta^2)(1+3\eta) \sin \frac{\pi}{3} \quad (4.280b)$$

die gegenüber (4.276a,b) eine bessere Approximation liefert, wie durch die kleinere Fehlernorm $L_2 = 0,0025$ im Vergleich zu $L_2 = 0,0097$ zum Ausdruck kommt. Die Fehlernorm ist mit Hilfe des folgenden MAPLE-Programms bestimmt worden.

>

⊙ 4.12-3.mws

```
>x(1,eta):=- (1/16)*(1-eta)*(1-9*eta^2)+(9/16)*(1-eta^2)*
(1-3*eta)*cos(Pi/6)+(9/16)*(1-eta^2)*(1+3*eta)*cos(Pi/3):
>y(1,eta):=- (1/16)*(1+eta)*(1-9*eta^2)+(9/16)*(1-eta^2)*
(1-3*eta)*sin(Pi/6)+(9/16)*(1-eta^2)*(1+3*eta)*sin(Pi/3):
>plot1:=plot([x(1,eta),y(1,eta), eta=-1..1], scaling=constrained, thickness=2):
>plot2:=plot(sqrt(1-x^2), x=0..1, scaling=constrained, thickness=1):
>plots[display]({plot1,plot2}):
>rr:=x(1,eta)^2+y(1,eta)^2:
>rr:=expand(%):
>rr:=factor(%):
>r:=+sqrt(%):
>L_2:=evalf(sqrt(int((1-r)^2,eta=-1..1))):
```

$L_{-2} := 0.002447333606$

```
>plot(1-r,eta=-1..1):
```

>

Darin können wiederum die einzelnen Zwischenergebnisse ausgedruckt werden, wenn man die Doppelpunkte durch Semikola ersetzt.

Zur Erzeugung *dreidimensionaler Elemente*, die krummflächig berandet sind, kann man unter Berücksichtigung einer dritten Funktion $z = z(\xi, \eta, \zeta)$ in (4.142) von einem *Master-Würfel* im ξ, η, ζ -Raum mit $-1 \leq (\xi, \eta, \zeta) \leq 1$ ausgehen (Bild 4.46 links). Zur Erläuterung der Vorgehensweise genügt es, eine Würfelfläche *isoparametrisch* auf eine gekrümmte Kontur abzubilden. Im Folgenden wird als Beispiel die Fläche $\xi = 1$ in Bild 4.46 links mit den Knotenpunkten

$$P_2^*(1, -1, -1), P_3^*(1, 1, -1), P_6^*(1, -1, 1), P_7^*(1, 1, 1),$$

$$P_{10}^*(1, 0, -1), P_{18}^*(1, 0, 1), P_{14}^*(1, -1, 0), P_{15}^*(1, 1, 0)$$

gewählt. Die zugehörigen Formfunktionen vom SERENDIPITY-Typ ergeben sich aus (4.253a,c,d) zu:

$$N_{2;3}(1, \eta, \zeta) = -\frac{1}{4}(1 \mp \eta)(1 - \zeta)(1 \pm \eta + \zeta), \quad (4.281a,b)$$

$$N_{6;7}(1, \eta, \zeta) = -\frac{1}{4}(1 \mp \eta)(1 + \zeta)(1 \pm \eta - \zeta), \quad (4.281c,d)$$

$$N_{10;18}(1, \eta, \zeta) = \frac{1}{2}(1 - \eta^2)(1 \mp \zeta), \quad N_{14;15}(1, \eta, \zeta) = \frac{1}{2}(1 - \zeta^2)(1 \mp \eta). \quad (4.281e-h)$$

Damit erhält man folgende *isoparametrische Abbildung* der Würfel­fläche $\xi = 1$ auf eine gekrümmte Fläche im (x, y, z) -Raum:

$$x(1, \eta, \zeta) = N_2(1, \eta, \zeta)x_2 + \dots + N_{15}(1, \eta, \zeta)x_{15}, \quad (4.282a)$$

$$y(1, \eta, \zeta) = N_2(1, \eta, \zeta)y_2 + \dots + N_{15}(1, \eta, \zeta)y_{15}, \quad (4.282b)$$

$$z(1, \eta, \zeta) = N_2(1, \eta, \zeta)z_2 + \dots + N_{15}(1, \eta, \zeta)z_{15}. \quad (4.282c)$$

Darin werden die Knotenpunkte $P_i(x_i, y_i, z_i)$ zur Anpassung an die Geometrie des Bauteils festgelegt. Als Beispiel werden folgende Knotenpunkte im (x, y, z) -Raum vorgegeben:

$$P_2(0, -1, -1), P_3(0, 1, -1), P_6(0, -1, 1), P_7(0, 1, 1),$$

$$P_{10}\left(\frac{1}{2}, 0, -1\right), P_{18}\left(\frac{1}{2}, 0, 1\right), P_{14}\left(\frac{1}{2}, -1, 0\right), P_{15}\left(\frac{1}{2}, 1, 0\right).$$

Bei diesem Beispiel werden die Eckpunkte ②, ③, ⑥, ⑦ nicht verschoben. Der Koordinatenursprung $(x = y = z = 0)$ liegt im Schwerpunkt der abzubildenden Würfel­fläche. Die isoparametrische Abbildung (4.282) ist mit Hilfe des 3D-MAPLE-Programms ausgewertet und in Bild 4.53 dargestellt.

>

⊙ 4.12-4.mws

> N[2]:=-(1/4)*(1-eta)*(1-zeta)*(1+eta+zeta):

> N[3]:=-(1/4)*(1+eta)*(1-zeta)*(1-eta+zeta):

> N[6]:=-(1/4)*(1-eta)*(1+zeta)*(1+eta-zeta):

> N[7]:=-(1/4)*(1+eta)*(1+zeta)*(1-eta-zeta):

> N[10]:=(1/2)*(1-eta^2)*(1-zeta):

> N[18]:=(1-eta^2)*(1+zeta)/2:

> N[14]:=(1-zeta^2)*(1-eta)/2:

```

> N[15]:= (1-zeta^2)*(1+eta)/2:
> x(eta,zeta):=(N[10]+N[18]+N[14]+N[15])/2:
> x(eta,zeta):=simplify(%):
> y(eta,zeta):=-N[2]+N[3]-N[6]+N[7]-N[14]+N[15]:
> y(eta,zeta):=simplify(%):
> z(eta,zeta):=-N[2]-N[3]+N[6]+N[7]-N[10]+N[18]:
> z(eta,zeta):=simplify(%):
> plot1:=plot3d([x(eta,zeta)/infinity,y(eta,zeta),z(eta,zeta)],eta=-1..1,
zeta=-1..1,axes=normal,orientation=[-60,80],style=wireframe):
> plot2:=plot3d([x(eta,zeta),y(eta,zeta),z(eta,zeta)],eta=-1..1,zeta=-1..1,
axes=normal,orientation=[-60,80],scaling=constrained,
style=patchcontour,shading=zhue,tickmarks=[3,3,3]):
> plots[display]({plot1,plot2}):
>

```




Bild 4.53 Isoparametrische Abbildung einer Würfel­fläche $\xi = 1$ auf eine konvex-gekrümmte Fläche

Die einzelnen Zwischenergebnisse können wiederum ausgedrückt werden, wenn man im MAPLE-Programm die Doppelpunkte durch Semikola ersetzt.

Für den Anwender sind auch *krummflächige Tetraederelemente* von grundlegender Bedeutung. Zur Generierung solcher Elemente bieten sich wiederum *isoparametrische Abbildungen* an. Als Grundaufgabe wird im Folgenden die *isoparametrische Abbildung* einer *Oktaederebene* im (ξ, η, ζ) -Raum mit den 6 Knotenpunkten (Bild 4.42)

$$P_2^*(1, 0, 0), P_3^*(0, 1, 0), P_4^*(0, 0, 1), P_6^*\left(\frac{1}{2}, \frac{1}{2}, 0\right), P_9^*\left(\frac{1}{2}, 0, \frac{1}{2}\right), P_{10}^*\left(0, \frac{1}{2}, \frac{1}{2}\right)$$

auf ein *sphärisches Dreieck* im (x, y, z) -Raum mit den Knotenpunkten

$$P_2(1, 0, 0), P_3(0, 1, 0), P_4(0, 0, 1),$$

$$P_6\left(\frac{1}{\sqrt{2}}, \frac{1}{\sqrt{2}}, 0\right), P_9\left(\frac{1}{\sqrt{2}}, 0, \frac{1}{\sqrt{2}}\right), P_{10}\left(0, \frac{1}{\sqrt{2}}, \frac{1}{\sqrt{2}}\right)$$

diskutiert. Dabei werden zur einfacheren Darstellung die Eckpunkte $P_1^*(0, 0, 0), \dots, P_4^*(0, 0, 1)$ des *Einheitstetraeders* (Bild 4.42) nicht verschoben. Die *isoparametrische Abbildung* erfolgt mit Hilfe der Formfunktionen (4.213) und (4.215a,b) gemäß

$$\begin{aligned} x(L_1 = 0, L_2, L_3, L_4) &= \sum_{i=1}^{10} N_i(L_1 = 0, L_2, L_3, L_4) x_i = \\ &= L_2(2L_2 - 1)x_2 + L_3(2L_3 - 1)x_3 + L_4(2L_4 - 1)x_4 \\ &\quad + 4L_2L_3x_6 + 4L_2L_4x_9 + 4L_3L_4x_{10}. \end{aligned} \quad (4.283)$$

Darin ist berücksichtigt, dass die Oktaederebene durch die natürliche Volumenordinate $L_1 = 0$ charakterisiert ist. Für das *Einheitstetraeder* bestehen zwischen den Volumenkoordinaten L_1, \dots, L_4 und den rechtwinkligen kartesischen Koordinaten ξ, η, ζ die einfachen Beziehungen (4.227a-d), so dass die *isoparametrische Abbildung* (4.283) gemäß

$$\begin{aligned} x(\xi, \eta, \zeta) &= \xi(2\xi - 1)x_2 + \eta(2\eta - 1)x_3 + \zeta(2\zeta - 1)x_4 \\ &\quad + 4\xi\eta x_6 + 4\xi\zeta x_9 + 4\eta\zeta x_{10} \end{aligned} \quad (4.284)$$

dargestellt wird. Darin kann ζ wegen $L_1 = 0$ und (4.227a) gemäß $\zeta = 1 - \xi - \eta$ eliminiert werden, so dass man schließlich mit den Koordinaten des *sphärischen Dreiecks* $P_2(1, 0, 0), \dots, P_{10}\left(0, \frac{1}{\sqrt{2}}, \frac{1}{\sqrt{2}}\right)$ folgende *isoparametrische Abbildung* erhält

$$x(\xi, \eta) = \xi(2\xi - 1) + 2\sqrt{2}\xi\eta + 2\sqrt{2}\xi(1 - \xi - \eta), \quad (4.285a)$$

$$y(\xi, \eta) = \eta(2\eta - 1) + 2\sqrt{2}\eta(1 - \xi - \eta) + 2\sqrt{2}\eta\xi, \quad (4.285b)$$

$$z(\xi, \eta) = (1 - \xi - \eta)(1 - 2\xi - 2\eta) + 2\sqrt{2}(1 - \xi - \eta)\xi + 2\sqrt{2}(1 - \xi - \eta)\eta \quad (4.285c)$$

Man erkennt, dass diese drei Gleichungen durch zyklische Vertauschungen,

$$\xi \rightarrow \eta \rightarrow \zeta = 1 - \xi - \eta \rightarrow \xi,$$

ineinander übergehen. Man kann sie kürzer gemäß

$$x = \xi(2\xi - 1) + 2\sqrt{2}(1 - \xi)\xi, \quad (4.285^* a)$$

$$y = \eta(2\eta - 1) + 2\sqrt{2}(1 - \eta)\eta, \quad (4.285^* b)$$

$$z = 1 - (3 - 2\sqrt{2})(\xi + \eta) - 2(\sqrt{2} - 1)(\xi + \eta)^2 \quad (4.285^* c)$$

darstellen, da sich einige Terme in (4.285) herauskürzen.

Mit Hilfe der 3D-Graphik des MAPLE-Programms ist die *isoparametrische Abbildung* (4.285^{*} a,b,c) in Bild 4.54a veranschaulicht. Außerdem ist die Oktaederebene $\xi + \eta + \zeta = 1$ im ersten Oktanten dargestellt. Zu beachten ist, dass zur Abbildung des sphärischen Dreiecks $\xi = 0..1$ **und** $\eta = 0..1 - \xi$ einzugeben ist (**plot1**), während zur Darstellung der Oktaederebene $\xi = 0..1$ **und** $\eta = 0..1$ zu berücksichtigen ist (**plot2**).

>

⊙ 4.12-5.mws

> x:=xi*(2*xi-1)+2*sqrt(2)*(1-xi)*xi;

$$x := \xi(2\xi - 1) + 2\sqrt{2}(1 - \xi)\xi$$

> y:=eta*(2*eta-1)+2*sqrt(2)*(1-eta)*eta;

$$y := \eta(2\eta - 1) + 2\sqrt{2}(1 - \eta)\eta$$

> z:=1-(3-2*sqrt(2))*(xi+eta)-2*(sqrt(2)-1)*(xi+eta)^2;

$$z := 1 - (3 - 2\sqrt{2})(\xi + \eta) - 2(\sqrt{2} - 1)(\xi + \eta)^2$$

> plot1:=plot3d([x,y,z],xi=0..1, eta=0..1-xi, axes=normal, orientation=[-60,80]):

> plot2:=plot3d(1-xi-eta,xi=0..1, eta=0..1, axes=normal, orientation=[-60,80],

view=[0..1,0..1,0..1],style=wireframe, shading=zhue,tickmarks=[3,3,3]):

> plots[display]({plot1,plot2});




Bild 4.54a Isoparametrische Abbildung der Oktaederebene auf ein sphärisches Dreieck mittels quadratischer Formfunktionen

```

> rr:=x^2+y^2+z^2:
> rr:=expand(%):
> rr:=simplify(%):
> r:=+sqrt(%):
> L_2]:=evalf(sqrt(int(int((1-r)^2,eta=0..1-xi),xi=0..1)));
Warning, computation interrupted

> l_2]:=evalf(sqrt(int(int((1-r^2)^2,eta=0..1-xi),xi=0..1)));
      l_2 := 0.08014965849

> delta:=(1-r^2)^2-(1-r)^2:
> plot3d(delta,xi=0..1,eta=0..1-xi,axes=normal,orientation=[-60,80]):
>

```

Im obigen MAPLE-Programm werden auch zwei *Fehlernormen* angegeben, die folgendermaßen definiert sind:

$$L_2 := \sqrt{\int_0^1 \int_0^{1-\xi} (1-r)^2 \, d\eta \, d\xi} \quad (4.286a)$$

$$\ell_2 := \sqrt{\int_0^1 \int_0^{\xi} \int_0^{\eta} (1-r^2)^2 d\eta d\xi} . \quad (4.286b)$$

Da MAPLE den ersten Ausdruck L_2 , d.h. die eigentliche L_2 -Fehlernorm, aufgrund des komplizierten Wurzelausdruckes r auch nach langer Rechenzeit nicht ausdrückte, wurde die Rechnung abgebrochen und eine modifizierte ℓ_2 -Fehlernorm eingeführt. Diesen Ausdruck kann MAPLE problemlos auswerten. Man erhält einen Wert von $\ell_2 = 0,08015$, der größer ist L_2 , da die Differenz $\Delta = (1-r^2)^2 - (1-r)^2 = r^4 - 3r^2 + 2r$ im gesamten Integrationsbereich positiv ist, wie die hier nicht ausgedruckte Graphik zeigt. Man kann sie ausdrucken lassen, wenn man im letzten Plot-Befehl den Doppelpunkt durch ein Semikolon ersetzt. Da der Unterschied Δ der Integranden im gesamten Integrationsgebiet sehr gering ist, kann die modifizierte ℓ_2 -Fehlernorm zur Diskussion herangezogen werden.

Eine *interaktive* numerische Auswertung der L_2 -Fehlernorm mit MAPLE zeigt folgender Ausdruck:

Schrittweite	Anz. Stützpunkte	L_2 -Norm
0.1	66	0.03616
0.01	5151	0.04105
0.005	20301	0.04135
0.001	501501	0.04159

Mit kleiner werdender Schrittweite ändert sich der L_2 -Wert nur geringfügig und konvergiert schließlich gegen einen Wert von $L_2 > 0,04159$, der günstiger ist als der gemäß (4.286b) ermittelte Wert von $\ell_2 = 0,08015$, wie oben vorausgesagt und in Tabelle 4.12 zum Ausdruck kommt. Die interaktive Auswertung mit einer Schrittweite von 0.001 benötigt eine enorm große Rechenzeit von etwa 20 min, während die GAUSS-Quadratur in einer extrem geringen Rechenzeit genauere Ergebnisse liefert

Zur Verbesserung der isoparametrischen Abbildung werden im Folgenden MAPLE-Programm *kubische Formfunktionen* aus (4.218) und (4.227a,b,c,d) mit $L_1 = 0$ eingesetzt. Gemäß Bild 4.39 werden 10 Knotenpunkte $P_i^*(\xi, \eta, \zeta)$ auf der Oktaederebene $L_1 = 0$ betrachtet,

$$P_2^*(1, 0, 0), P_3^*(0, 1, 0), P_4^*(0, 0, 1), P_7^*\left(\frac{2}{3}, \frac{1}{3}, 0\right), P_8^*\left(\frac{1}{3}, \frac{2}{3}, 0\right), \\ P_{13}^*\left(\frac{2}{3}, 0, \frac{1}{3}\right), P_{14}^*\left(\frac{1}{3}, 0, \frac{2}{3}\right), P_{15}^*\left(0, \frac{2}{3}, \frac{1}{3}\right), P_{16}^*\left(0, \frac{1}{3}, \frac{2}{3}\right), P_{19}^*\left(\frac{1}{3}, \frac{1}{3}, \frac{1}{3}\right),$$

und auf Punkte $P_i(x, y, z)$ der Einheitskugel verschoben:

$$P_2(1, 0, 0), P_3(0, 1, 0), P_4(0, 0, 1), P_7\left(\cos \frac{\pi}{6}, \sin \frac{\pi}{6}, 0\right),$$

$$P_8 \left(\cos \frac{\pi}{3}, \sin \frac{\pi}{3}, 0 \right), P_{13} \left(\cos \frac{\pi}{6}, 0, \sin \frac{\pi}{6} \right), P_{14} \left(\cos \frac{\pi}{3}, 0, \sin \frac{\pi}{3} \right),$$

$$P_{15} \left(0, \cos \frac{\pi}{6}, \sin \frac{\pi}{6} \right), P_{16} \left(0, \cos \frac{\pi}{3}, \sin \frac{\pi}{3} \right), P_{19} \left(\frac{1}{\sqrt{3}}, \frac{1}{\sqrt{3}}, \frac{1}{\sqrt{3}} \right),$$

Damit ergibt sich die *isoparametrische Abbildung* in Erweiterung von (4.283) bzw. (4.285) zu:

$$x = N_2 + (N_7 + N_{13}) \cos \frac{\pi}{6} + (N_8 + N_{14}) \cos \frac{\pi}{3} + N_{19} / \sqrt{3}, \quad (4.287a)$$

$$y = N_3 + N_7 \sin \frac{\pi}{6} + N_8 \sin \frac{\pi}{3} + N_{15} \cos \frac{\pi}{6} + N_{16} \cos \frac{\pi}{3} + N_{19} / \sqrt{3}, \quad (4.287b)$$

$$z = N_4 + (N_{13} + N_{15}) \sin \frac{\pi}{6} + (N_{14} + N_{16}) \sin \frac{\pi}{3} + N_{19} / \sqrt{3}. \quad (4.287c)$$

Darin ermittelt man die kubischen Formfunktionen aus (4.218) und (4.227a,b,c,d) mit $L_1 = 0$ in Abhängigkeit der Koordinaten ξ und η , die im Folgenden MAPLE-Programm als Parameter fungieren (*parametric 3d-plot*).

⊙ 4.12-6.mws

```
>
> N[2]:=xi*(3*xi-1)*(3*xi-2)/2: N[3]:=eta*(3*eta-1)*(3*eta-2)/2:
> N[4]:=(1-xi-eta)*(3*(1-xi-eta)-1)*(3*(1-xi-eta)-2)/2:
> N[7]:=9*(3*xi-1)*xi*eta/2: N[8]:=9*(3*eta-1)*xi*eta/2:
> N[13]:=9*(3*xi-1)*xi*(1-xi-eta)/2:
> N[14]:=9*(3*(1-xi-eta)-1)*xi*(1-xi-eta)/2:
> N[15]:=9*(3*eta-1)*eta*(1-xi-eta)/2:
> N[16]:=9*(3*(1-xi-eta)-1)*eta*(1-xi-eta)/2:
> N[19]:=27*xi*eta*(1-xi-eta):
> x:=N[2]+(N[7]+N[13])*cos(Pi/6)+(N[8]+N[14])*cos(Pi/3)+N[19]/sqrt(3):
> x:=expand(%):
> x:=simplify(%):
> x:=factor(%):
> y:=N[3]+N[7]*sin(Pi/6)+N[8]*sin(Pi/3)+N[15]*cos(Pi/6)+N[16]*cos(Pi/3)
  +N[19]/sqrt(3):
> y:=expand(%):
> y:=simplify(%):
> y:=factor(%):
> z:=N[4]+(N[13]+N[15])*sin(Pi/6)+(N[14]+N[16])*sin(Pi/3)+N[19]/sqrt(3):
> z:=expand(%):
> z:=simplify(%):
> z:=factor(%):
> plot1:=plot3d([x,y,z],xi=0..1,eta=0..1-xi,axes=normal,
  orientation=[-60,80],tickmarks=[2,2,2]):
```

```

> plot2:=plot3d(1-xi-eta,xi=0..1,eta=0..1,axes=normal,orientation=[-60,80],
view=[0..1,0..1,0..1],style=wireframe,shading=zhue,tickmarks=[2,2,2]):
> plots[display]({plot1,plot2});
>

```




Bild 4.54b Isoparametrische Abbildung der Oktaederebene auf ein sphärisches Dreieck mittels kubischer Formfunktionen

```

> rr:=x^2+y^2+z^2:
> rr:=expand(%):
> rr:=simplify(%):
> r:=+sqrt(%):
> L_2]:=evalf(sqrt(int(int((1-r)^2,eta=0..1-xi),xi=0..1)));

```

$L_{-2} := 0.001185186860$

```

> l_2]:=evalf(sqrt(int(int((1-r^2)^2,eta=0..1-xi),xi=0..1)));

```

$l_{-2} := 0.002368576825$

```

> delta:=(1-r^2)^2-(1-r)^2:
> plot3d(1-r,xi=0..1,eta=0..1-xi,axes=normal,
orientation=[-60,80],tickmarks=[3,3,3]);

```




Bild 4.55 Abstand $(1-r)$ der isoparametrischen Abbildung vom Einheitskreis; Näherung mittels kubischer Formfunktionen

Zwischen der quadratischen (Bild 4.54a) und kubischen (Bild 4.54b) isoparametrischen Abbildung ist optisch kaum ein Unterschied zu erkennen. Die *Fehlernormen*

Formfunktionen	$L_2 - (4.286a)$	$\ell_2 - (4.286b)$
quadratisch	0,041598	0,080149
kubisch	0,001185	0,002368

sind jedoch bei der kubischen Näherung wesentlich geringer als bei der quadratischen. Die Abstandsfunktion $(1-r)$ der kubischen Näherung vom Einheitskreis ist in Bild 4.55 dargestellt. Deren „quadratische Mittelung“ gemäß Definition (4.286a) führt auf den niedrigen Wert $L_2 = 0,0011852$. Die „modifizierten“ Größen ℓ_2 gemäß Definition (4.286b) sind größer als die L_2 -Werte, wie bereits oben vermerkt.

Zu bemerken ist, dass MAPLE die L_2 -Fehlernorm der kubischen Näherung problemlos nach (4.286a) ausgewertet hat, während die quadratische Näherung bei der Integration nach (4.286a) auf Schwierigkeiten stieß, wie bereits oben erwähnt.

Zur *numerischen Integration* bietet MAPLE verschiedene klassische Verfahren an. Falls keine Angaben in den Optionen gemacht werden, benutzt MAPLE im

Allgemeines das *CLENSHAW-CURTIS-Verfahren* und weicht bei „Schwierigkeiten“ auf die *NEWTON-COTES-Integration* (`_NCrule`) oder das adaptive *doppelt-exponentielle Verfahren* (`_Dexp`) aus. Zur Umgehung der von MAPLE bevorzugten internen Integration (`_CCquad`) können die beiden anderen Optionen (`_NCrule` oder `_Dexp`) auch direkt im Programm angesteuert werden.

Zur Ermittlung der L_2 -Norm nach (4.286a) unter Berücksichtigung quadratischer Formfunktionen haben jedoch die in MAPLE implementierten Verfahren (vermutlich aufgrund der auftretenden speziellen elliptischen Integrale) versagt.

Daher wurde zunächst eine interaktive Aufsummierung über eine Vielzahl von Stützwerten des Integranden $(1-\eta)^2$ bei entsprechender Schrittweite benutzt, wie oben erwähnt. Diese Vorgehensweise erfordert jedoch eine enorm große Rechenzeit. Im Folgenden soll daher die klassische *GAUSS-Quadratur* eingesetzt werden, die bereits bei einer geringen Anzahl von Integrationsstützpunkten Ergebnisse mit hoher Genauigkeit liefert und im Allgemeinen der *NEWTON-COTES-Integration* überlegen ist. In der NC-Quadratur können bei hohen Polynomgraden negative Wichtungsfaktoren erscheinen, was eine *Instabilität des Verfahrens* zur Folge haben kann, d.h., durch „Stellenauslöschung“ bewirken kleine Änderungen im Integranden stärkere Änderungen im ermittelten Näherungswert als im Integralwert.

Zur numerischen Ermittlung der L_2 -Fehlernormen (4.286a,b) der isoparametrischen Abbildungen in Bild 4.54a,b bietet sich die *GAUSS-Quadratur* gemäß

$$\int_0^1 \int_0^{1-\xi} f(\xi, \eta) d\eta d\xi \approx \sum_{i=1}^n w_i f(\xi_i, \eta_i) \quad (4.288)$$

an. Darin sind n die Anzahl der vorgegebenen *Integrationsstützstellen* (ξ_i, η_i) und w_i die zugehörigen *Wichtungsfaktoren*, die man den Tabellen 4.10a,b für $n=7$ und $n=13$ entnehmen kann.

Zu bemerken ist, dass die in den Tabellen 4.10a,b aufgelisteten Parameter nicht nur zur Integration über das *Einheitsdreieck*, sondern auch für ein finites Dreieck in allgemeiner Lage verwendet werden können, wenn man Dreieckskoordinaten L_1, L_2, L_3 einführt, wie in den Tabellen angedeutet.

Wegen $L_1 + L_2 + L_3 = 1$ bestätigt man für die Tabellenwerte: $a + b + a = c + d + c$ usw. in Tabelle 4.10a oder $a + a + b = e + f + g$ usw. in Tabelle 4.10b. Für die *Wichtungsfaktoren* muss gelten: $(w_s + 3w_1 + 3w_4)/A_\Delta = 1$ in Tabelle 4.10a und $(w_s + 3w_1 + 3w_4 + 6w_7)/A_\Delta = 1$ in Tabelle 4.10b, da

$$A_\Delta = \iint_{A_\Delta} (1) dA \approx \sum_{i=1}^n (1)_i w_i = \begin{cases} w_s + 3w_1 + 3w_4 & \text{(Tabelle 4.10a)} \\ w_s + 3w_1 + 3w_4 + 6w_7 & \text{(Tabelle 4.10b)} \end{cases}$$

gilt. Die angegebenen *Wichtungsfaktoren* erfüllen diese Forderung.

Tabelle 4.10b GAUSS-Quadratur auf dem Einheitsdreieck mit **n = 13** Stützstellen

	$a = 0,26034\ 59661$ $b = 0,47930\ 80678$ $c = 0,06513\ 0103$ $d = 0,86973\ 97942$ $e = 0,63844\ 41886$ $f = 0,31286\ 5496$ $g = 0,04869\ 03154$ <hr style="border-top: 1px dashed black;"/> $w_S / A_\Delta = -0,14957005$ $w_1 / A_\Delta = 0,17562$ $w_4 / A_\Delta = 0,05334724$ $w_7 / A_\Delta = 0,077113761$ <hr style="border-top: 1px dashed black;"/> $L_1 = 1 - \xi - \eta$ $L_2 = \xi$ $L_3 = \eta$		
GAUSS-Punkte	Koordinaten (L_1, L_2, L_3)	Koordinaten (ξ, η)	Wichtungsfaktoren (w_i)
S	(1/3, 1/3, 1/3)	(1/3, 1/3)	w_S
1	(b, a, a)	(a, a)	w_1
2	(a, b, a)	(b, a)	$w_2 \equiv w_1$
3	(a, a, b)	(a, b)	$w_3 \equiv w_1$
4	(d, c, c)	(c, c)	w_4
5	(c, d, c)	(d, c)	$w_5 \equiv w_4$
6	(c, c, d)	(c, d)	$w_6 \equiv w_4$
7	(g, e, f)	(e, f)	w_7
8	(g, f, e)	(f, e)	$w_8 \equiv w_7$
9	(f, g, e)	(g, e)	$w_9 \equiv w_7$
10	(e, g, f)	(g, f)	$w_{10} \equiv w_7$
11	(e, f, g)	(f, g)	$w_{11} \equiv w_7$
12	(f, e, g)	(e, g)	$w_{12} \equiv w_7$

Tabelle 4.10a GAUSS-Quadratur auf dem Einheitsdreieck mit $n = 7$ Stützstellen

$$a = (6 + \sqrt{15})/21$$

$$b = (9 - 2\sqrt{15})/21$$

$$c = (6 - \sqrt{15})/21$$

$$d = (9 + 2\sqrt{15})/21$$

$$w_S / A_\Delta = 0.225$$

$$w_1 / A_\Delta = (155 + \sqrt{15})/1200$$

$$w_4 / A_\Delta = (155 - \sqrt{15})/1200$$

$$A_\Delta = 1/2$$

$$L_1 = 1 - \xi - \eta$$

$$L_2 = \xi$$

$$L_3 = \eta$$

GAUSS-Punkte	Koordinaten (L_1, L_2, L_3)	Koordinaten (ξ, η)	Wichtungsfaktoren (w_i)
S	(1/3, 1/3, 1/3)	(1/3, 1/3)	w_S
1	(b, a, a)	(a, a)	w_1
2	(a, b, a)	(b, a)	$w_2 = w_1$
3	(a, a, b)	(a, b)	$w_3 = w_1$
4	(d, c, c)	(c, c)	w_4
5	(c, d, c)	(d, c)	$w_5 = w_4$
6	(c, c, d)	(c, d)	$w_6 = w_4$

Die GAUSS-Quadratur liefert exakte Werte für Polynomterme $x^p y^q$ mit $p+q \leq m$, wobei $m = 5$ bei einer Quadratur mit $n = 7$ GAUSS-Punkten (Tabelle 4.10a) und $m = 7$ bei $n = 13$ GAUSS-Punkten (Tabelle 4.10b). Dazu sei das Integral

$$\int_0^{1-x} \int_0^x x^p y^q dy dx = \frac{\Gamma(p+1)\Gamma(q+1)}{\Gamma(p+q+3)} = \frac{p!q!}{(p+q+2)!} \tag{4.289}$$

mit $p \geq 0, q \geq 0$ aus Ü 3.1.31 mittels der GAUSS-Quadratur (4.288) angenähert.

Die numerischen Ergebnisse sind in Tabelle 4.11 mit den exakten Werten verglichen.

Tabelle 4.11 GAUSS-Quadratur des Integrals (4.289) im Vergleich mit den exakten Werten

Polynomgrad $p+q=m$	Exakte Werte aus (4.289)	GAUSS-Quadratur (4.288)	
		$n=7$	$n=13$
5	$1/420$ ≈ 0.00238095	exakt	exakt
6	$1/840$ ≈ 0.0011905	0.0011464	exakt
7	$1/2520$ ≈ 0.0003968	0.00041812	exakt
8	$1/5040$ ≈ 0.000198413	0.00019692	0.000200726

Der Integrand in der ℓ_2 -Norm (4.286b) enthält Polynomterme maximal vom Grade $m=6$, deren Quadratur mit $n=13$ gemäß Tabelle 4.11 exakte Werte liefert.

In Tabelle 4.12 sind Fehlernormen (4.286a,b), die nach verschiedenen Integrationsverfahren ermittelt wurden, zusammengestellt, wobei die *GAUSS-Quadratur* mit $n=13$ wohl die genauesten Werte liefert.

Tabelle 4.12 Fehlernormen L_2 und ℓ_2 gemäß (4.286a,b) der isoparametrischen Abbildungen (4.285) und (4.287)

			quadratisch Formfunktionen	kubische Formfunktionen
MAPLE V8	L_2	--)* --		0.0011852
	ℓ_2		0.0801497	0.0023686
Interaktive Aufsummierung (Schrittweite = 0.001)	L_2		0.0415987	0.0011853
	ℓ_2		0.0800313	0.0023688
GAUSS Quadratur	$n=7$	L_2	0.0416136	0.0014019
		ℓ_2	0.0797572	0.0027998
	$n=13$	L_2	0.0416464	0.0010104
		ℓ_2	0.0800670	0.0020199

)* Der Wert konnte mit MAPLE nicht berechnet werden

Obige grundlegende Untersuchungen zu isoparametrischen Abbildungen zeigen, dass derartige Abbildungen sehr geeignet sind, räumliche Elemente zu erzeugen und vorgegebene Konfigurationen mit hoher Genauigkeit zu approximieren. Hierzu sind einige Beispiele in Bild 4.56 zusammengestellt.




Bild 4.56 Isoparametrische räumliche Elemente

Die in dieser Ziffer diskutierten Grundlagen isoparametrischer Abbildungen mögen den Anwender bei der Erzeugung eigener 3D-Elemente unterstützen, die in kommerziellen Programmen für seine Belange nicht zu finden sind.